

# Aircraft Search Utility

by Claudio "Cloudy" Di Veroli, published in PC FLIGHT,  
The Official magazine of the "PC Pilots Ireland", Sep. 2014, pp. 43-46

Recent versions of Microsoft Flight Simulator® install by default a few dozen aircraft. Adding a few freeware and payware ones to your flight simulator installation, it is not uncommon to have in the region of 50 aircraft (like myself), not counting variants (such as different liveries and panels). Some flightsimmers have 100 or more. At this point, when you are not running your flight simulator, you often need to access a particular aircraft: perhaps to check or edit a file in the aircraft folder, such as the aircraft configuration file, the checklist or the panel configuration inside the panel subfolder. Or else you wish to see the aircraft external looks or a "screenie" of the instrument panel, stored perhaps in a screenies folder. And there you go, navigating Windows Explorer until you find folder and file. Cumbersome, especially if you also have there hundreds of AI aircraft installed by a Traffic addon: you may easily need one full minute to get there! Solution: I have two different ways to have access any of my aircraft with just a few keypresses. Let me share them with you.

## Microsoft Access

Needless to say, if you have your flightsim software and aircraft information neatly stored into a relational database, things could not be easier. My flightsim database is started with a shortcut associated to the keypress F12. Once you are in the database, the Aircraft form starts by pressing F4: for this you define a Macro with the name "AutoKeys" and inside it a Macro with "Macro Name" {F4} and Action "OpenForm", for which you store Form Name "Aircraft".



Once in the Aircraft form, I have a search box sorted by aircraft Maker and Model, or else I can just press Control-F and enter any string to locate an aircraft record. Once I am there, the field ("control" in Access jargon) "FS Folder" contains the aircraft's folder name in my flight simulator.

Near to it you can see a button "Expl...", associated with the following Visual Basic code:

```
Private Sub Explorer_Click()  
    Dim AirFol  
    AirFol = "S:\FS2004\Aircraft\  
    Shell ("Explorer.exe " & AirFol & Me.FS_Folder)  
End Sub
```

So let us say you wish to open a file inside your Beech Baron 58 folder. The sequence is as follows:

- |  |   |
|--|---|
| 1-Press F12                                    | (opens the Access database)             |
| 2-Press F4                                     | (opens the Aircraft form)               |
| 3-Press Control-F                              | (opens the Find dialog)                 |
| 4-Type a substring of the aircraft folder name |   |
| 5-Press Enter                                  | (shows the aircraft record in the form) |
| 6-Click the "Explorer" button                  | (opens the folder in Explorer).         |

This is very neat, but only practical if you are already inside the Aircraft form. If you are not, it is cumbersome and not too fast: I have timed the above 6 steps in a very fast computer: 18 seconds. Then, again, most flightsimmers do not have such a database to begin with. So let us find a more general and faster solution.

## **VBScript utility**

This will work in any Windows PC with VBScript enabled (some users disable it for security, but this is unnecessary: VBScript is very useful for lots of tasks, and any security issues it might produce are very well catered for by any of the common freeware antivirus). There are two provisos:

- Every aircraft folder name should not contain any spaces: use underscore symbols "\_" instead. These changes will not invalidate the aircraft installation.
- If you have an AI Traffic addon (they all install hundreds of aircraft inside your main Aircraft folder) you need a way to filter out these folders. For this, and also to make any search faster, we first need to have within the Aircraft folder a simple text file containing the list of aircraft folder names. To produce this file you just open a command box in the Aircraft folder and issue

- `dir *.* /b > AircraftFolders.txt`

You should then delete from the file the AI aircraft folders. Let me now show you the VBScript code for a lightning-fast aircraft search. The code is stored in a simple text file with extension ".vbs". You will just need to make minimal amendments (for which no coding knowledge is necessary) to adapt it to your flight simulator installation. It is also assumed that you have a folder containing aircraft external views, another folder with panel screenshots, and in both cases these are pictures with extension JPG and file name identical to the aircraft folder name. The code shown below has been structured for minimal verbosity. It consists of three sections.

## Folder search code

This is the first part of the code. It starts with usual initial settings, including strings that the user can change to adapt the code to different folder paths. After asking the user to enter a string, the code searches for it in Aircraft.Folders.txt. If it is not found, since searches in VBScript are case-sensitive, the string is converted to lower case and the search is repeated. If not found, the first letter is converted to upper case. If not found, the whole string is converted to upper case.

```
Dim wss, fso, ans, ant, carr, car2, cas, fil, foc, fol, got, irf, mas, nuc, pos, pic, wex
Set wss = WScript.CreateObject("WScript.Shell")
Set fso = CreateObject("Scripting.FileSystemObject")
carr = Chr(13) & Chr(10) : car2 = carr & carr
'
mas = "S:\Fs2004\Aircraft"           'path to aircraft folders
pic = "D:\Docs\FlightSim\Pictures\"  'path to your main FS pictures folder
irf = "F:\Utilities\IrfanView\i_view32.exe " 'path to your picture viewer followed by a space
wex = "explorer.exe " 'path (if needed) and executable to Windows Explorer or other file manager
'
ans = InputBox("Type any part of an aircraft folder name:","Aircraft menu - Folder Search")
if ans="" then WScript.Quit
'
' find fol, the first folder-with-no-path where the string ans is included
got = 0 : Call Sea("T")           'search for string as typed
if got = 0 then Call Sea("L")     'search for string all in lower case
if got = 0 then Call Sea("I")     'search with first letter in upper case rest in lower case
if got = 0 then Call Sea("U")     'search for string all in Upper case
if got = 0 then msgbox "String '" & ans & "' not found." : WScript.Quit
```

## Search subroutine

This is the third part of the code, located at the end of the .vbs file. The same subroutine performs differently according to the value of the *cat* parameter.

```
Sub Sea(cat)
Set fil = fso.OpenTextFile("s:\Fs2004\Aircraft\AircraftFolders.txt",1)
Do While Not fil.AtEndOfStream
    lin = fil.ReadLine           'reads a folder full name
    Select Case cat
        Case "T" : cas = ans
        Case "L" : cas = Lcase(ans)
        Case "I" : cas = UCase(Left(ans,1)) & LCase(Right(ans,Len(ans)-1))
        Case "U" : cas = Ucase(ans)
    End Select
    pos = Instr(lin,cas)         'searches for string cas in folder name lin
```

```

        If pos<>0 then                'if pos is not zero, the string has been found
            fol = lin : got = 1
            exit do
            end if
    Loop
fil.Close
End Sub
'-----

```

If the search performed by the Subroutine *Sea* is successful, *fol* will contain the full aircraft folder name, and *got* will have the value 1.

### Action code

This is the second part of the main code, located immediately after the above "Folder search code".

```

'                ask for desired action
foc = Left(fol,12)        'if needed display only first 12 characters of folder
ant = InputBox("Please select action on aircraft " & fol & car2 & _
    "A  Explorer on ..\Aircraft\" & foc & "...\" & car2 & _
    "P  Explorer on ..\Aircraft\" & foc & "...\Panel\" & car2 & _
    "M  View Screenie in ..\Pictures ... Models\" & car2 & _
    "S  View Screenie in ..\Pictures ... Panels\" & car2 & _
    "Q  Quit (cancel)","Aircraft menu - Action Select","A")
'                actions on the searched aircraft:
ant = UCase(ant)        'allows Select Case to work even if lower case letter is entered
Select Case ant
    Case "A" : wss.run wex & mas & fol
    Case "P" : wss.run wex & mas & fol & "\panel"
    Case "M" : wss.run irf & pic & "Models\*" & ans & "*.JPG"
    Case "S" : wss.run irf & pic & "Panels\*" & ans & "*.JPG"
    Case "Q" : WScript.Quit
    Case "" : WScript.Quit
    Case Else : tem = wss.Popup("Invalid argument. Process aborted",2,fol,64) : WScript.Quit
End Select
'-----

```

## Customisation

Download Aircraft\_search.vbs from [http://cloudy.braybaroque.ie/cloudy\\_files/](http://cloudy.braybaroque.ie/cloudy_files/) . Open it with Notepad, check the text and edit the paths as needed for your flightsim installation. Associate this .vbs file with a shortcut and a keystroke, for example Alt-F12.


## Using the script

Once the script has been suitably customized, its use could not be easier. Let us assume that we wish to search for the aircraft in folder "Cessna\_C182RG\_II\_Skylane-Carenado" but we do not remember the aircraft code, just the "lane" thing. The sequence is as follows:

- 1-Press Alt-F12 (shows the "Aircraft menu - Folder Search" VBScript dialog)
- 2-Type the substring ("lane" say)




- 3-Press Enter (shows the "Aircraft menu - Action Select" VBScript dialog)
- 4-Type a letter for the action desired (e.g. "P" to go to the aircraft panel folder)



5-Press Enter again

(shows the desired folder or picture)



I have timed the whole procedure: 5 seconds only! This is orders of magnitude faster than the manual or database methods we saw above. It is even more practical when you wish to compare features of two or more different aircraft and need their folders or files open simultaneously.

[The following was added after publication on PC FLIGHT)

To conclude, let me comment that VBScript code has proven an invaluable tool in flight simulation file management. Find below a list of other scripts I have developed that make my everyday flightsimmer life easier. These are shown for FS2004, but they either also work or can be easily adapted for other flight simulators.

**Aircraft\_Taxonomy.** This script reads every "title=..." line of every aircraft.cfg, and produces an updated full list of installed aircraft titles, AircraftTitles.txt. This file is used by the Flight Process utility below.

**Cfg\_Msg.** Displays in a box some basic FS configuration settings (stored in fs9.cfg in FS2004) such as Extended Textures, Shadows, Vertex value and Frame rate limit.

**Delete\_0001.** Deletes gauges ending in 0001. These are generated by FS2004 and may be needed during flight. Once FS is shut down, however, they persist and accumulate, generating unnecessary file duplication.

**Flight\_Process.** When double-clicking on a flight .FLT or .WX file, instead of directly opening it in Flight Simulator, a dialog appears: pressing R allows to rename both files, pressing A opens a list of aircraft to select one and change the flight's aircraft before starting FS, and finally pressing F runs FS. A diagnostic line is displayed which tells you whether or not the aircraft referred to inside the .FLT is indeed installed in FS: for this it checks the AircraftTitles.txt obtained by the Aircraft\_Taxonomy script. If it is not, a warning is issued to prevent you starting FS. If you start FS anyway by pressing F, the script will not run, issuing an error message. (This avoids the well-known condition whereby, if you try to start FS2004 from a flight with a non-existent aircraft, FS2004 hangs in a nasty way, with error messages and a black screen that requires some time just to shut down the application).

**G13\_Kill\_Restart.** Kills and restarts the Logitech G13 gamepad in case the Profile utility has stopped working (luckily a rare occurrence). A similar code can be written for most USB hardware, and even set to run automatically after the PC comes back from "sleep". In the author's PC this is needed for both the CH Pro Pedals and the Audigy 2ZS sound card.

**Night\_Scenery.** There is a wealth of VFR photographic freeware sceneries to be downloaded. Unfortunately all of them have day textures only and are incompatible with autogen. As a consequence, during night flight outside airports, you see a black surface: no city lights, no road lights! Certainly at night time we could not care less about photographic textures and would like instead to see the "normal" night lights! This utility assumes that the user has manually set up a Scenery.txt file with the list of all the VFR sceneries. When the utility runs, a dialog lets the user select between "Night" (whereby scenery.cfg is edited to ensure that all the VFR sceneries listed in Scenery.txt have the parameter Active=FALSE) or "Day" (whereby they all have the parameter Active=TRUE).

